NASA TECHNICAL NOTE

NASA TN D-7443

NASA TN D-7443

FLUTTER OF PANELS ON DISCRETE FLEXIBLE SUPPORTS

by Manuel Stein

Langley Research Center
Hampton, Va. 23665

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • MARCH 1974

FLUTTER OF PANELS ON DISCRETE FLEXIBLE SUPPORTS

By Manuel Stein
Langley Research Center

SUMMARY

The supersonic flutter of wide panels on discrete flexible supports is investigated for three different panel-support configurations. The first study examines the effect of support stiffness on the flutter behavior of a specific five-support configuration with leading- and trailing-edge overhangs; this configuration was recently under consideration for a heat shield on an entry body. The second study investigates the effect of support stiffness on flutter of panels with various numbers of equally spaced supports. The third study examines the effect of center-support location on the flutter of panels with three supports. Results are presented in nondimensional form. The analysis is based on wide-plate structural theory and Ackeret aerodynamics. Finite differences are employed to obtain solutions for flutter pressure. A computer program based on this analysis and including a direct solution technique is presented. The program can be used to find the flutter pressure of wide panels of variable thickness supported by any number of flexible supports.

INTRODUCTION

Panels held in place with rows of standoff supports, as shown in figure 1, have been considered for thermal protection systems of vehicles designed to withstand reentry from space. Static design of such systems generally leads to panels of minimum gage that are held in place by rather flexible supporting structure. However, dynamic considerations, particularly panel flutter, must also be satisfied and may dictate panel gage as well as dimensions of the supporting structure. Panel flutter may also be a design consideration in aircraft and other vehicles in which the skin panels often have lines of discrete supports.

A general discussion of flutter of panels is presented in reference 1. Results for the flutter of wide panels on nondeflecting supports are presented in reference 2; however, no analytical results are available for the flutter of panels on multiple discrete flexible supports. Wide-plate structural theory coupled with Ackeret aerodynamic theory can be used to obtain meaningful estimates of supersonic flutter speeds (dynamic pressure) for many configurations of interest with a minimum amount of computation. Additionally, discrete flexible supports can be readily accounted for through the method of writing equilibrium in terms of finite differences. Thus, in order to provide flutter predictions for wide panels on multiple flexible supports, equilibrium equations were developed in conjunction with this combination of theories (see "Method of Analysis" and appendix A), and a computer program was written which employs a direct solution technique rather than the usual modal approach. A listing of this program, which can be used for a wide class of flutter problems, is presented in appendix B.

This paper presents three parameter studies obtained by use of the computer program. The first study examines the effect of support stiffness on flutter behavior of panels in the configuration of figure 1. In the second study, the effect of support stiffness on flutter of panels with various numbers of equally spaced supports is examined. In the third study, the effect of the location of the center support of a three-support panel is examined.

SYMBOLS

a	length of panel
C	support rotational stiffness
c	speed of sound
D	plate bending stiffness
j,j_n,n	integers
k	support stiffness per unit width
M	Mach number
$M_x,\overline{M}_x$	bending moments per unit width (see eqs. (A5) and (A2))
m	mass per unit area
N	number of rows of supports
q	dynamic pressure

t	time
w,W	deflections of panel
x	coordinate in lengthwise direction
$\beta = \sqrt{M^2 - 1}$	
δ_{jj_n}	Kronecker delta
$\delta(x-x_n)$	Dirac delta function
ϵ	spacing of nodal points
ρ	air density
ω	frequency

Primes denote derivatives with respect to x.

METHOD OF ANALYSIS

The structure considered is a wide plate with the length interrupted by lines of spring supports (fig. 1). The plate is in a supersonic airstream with airflow perpendicular to the lines of support. The loading considered includes inertia and aerodynamic loading appropriate for flutter at Mach numbers of about 1.4 and higher. (See ref. 2.) In this section of the paper, terms in the differential equation of equilibrium will be identified, and the method of solution will be indicated.

The differential equation of equilibrium of forces for a wide panel on N supports is

$$\frac{\partial^2 \overline{M}_x}{\partial x^2} + \sum_{n=1}^{N} \left(kw - C \frac{\partial^2 w}{\partial x^2}\right) \delta(x-x_n) + m \frac{\partial^2 w}{\partial t^2} + \frac{2q}{\beta} \frac{\partial w}{\partial x} + \rho c \frac{\partial w}{\partial t} = 0 \tag{1}$$

where $\overline{M}_x = D \frac{\partial^2 w}{\partial x^2}$. The first term of equation (1) represents the restoring forces due to the bending stiffness D of the panel. The terms in the summation represent the restoring forces due to the N rows of springs, each having an extensional stiffness k and a rotational stiffness C. The next term represents the inertia force. The other terms are the aerodynamic lift associated with the local angle of attack $\frac{\partial w}{\partial x}$ and the time rate of change in deflection $\frac{\partial w}{\partial t}$. Both these terms come from quasi-steady Ackeret theory (or piston theory with compressibility correction). Aerodynamic damping effects, which are represented by the last term, are expected to be small for most cases (ref. 3).

For flutter, periodic motion may be assumed and equation (1) is thus converted to an ordinary differential equation. This equation together with the boundary conditions is put in finite-difference form by defining nodal points and by using conventional central differences. The stiffnesses and the mass were considered to be functions of x, which introduces no important complications with the use of finite differences. Thus, solution, of variable-cross-section beams supported by various springs is permitted. The details of the derivation are presented in appendix A, which also includes the method of solution and in appendix B, the computer program that was used in obtaining the results of the parameter studies in this paper. The method of solution used is direct, not modal, and requires the evaluation of a second-order determinant independent of the number of unknowns in the problem; the method then makes use of determinant plotting to determine the flutter pressure.

PROBLEMS STUDIED

All the configurations studied are wide panels and the airflow is along the length of the panel.

Special Five-Support Configuration

The first problem considered is the determination of the flutter pressure of a panel with five equally spaced supports and with overhangs like the panel of figure 1. The leading overhang is 10.7 percent of the support spacing and the trailing overhang is two-thirds of the leading overhang. The supports are flexible in extension but free to rotate. Flutter pressure is determined for a full range of support stiffness.

Panels On Equally Spaced Supports

The second problem considered is the determination of the flutter pressure of panels with two, three, four, or five equally spaced supports, where the outer supports are at the leading and trailing edges of the panel. Again, the supports are flexible in extension but free to rotate and flutter pressure is determined for a full range of support stiffness. Comparison of the results for the five-support case with the results for the special configuration indicates the effect of the overhangs on the flutter pressure.

Panels On Unequally Spaced Supports

A third set of problems considered is the flutter of panels with simple supports at the ends and with an additional simple support at various locations. Here the flutter pressure is determined as a function of the location of the additional support.

RESULTS AND DISCUSSION

All the problems were analyzed in terms of nondimensional parameters so that they apply to a wide class of dimensions within the specified configuration.

Special Five-Support Configuration

Vibration frequencies were determined for the configuration of figure 1 and the first 11 natural frequencies are plotted in figure 2 as a function of support stiffness. The natural frequencies group together in fours, and pairs within each group appear to touch. As the support stiffness increases, the frequencies increase until they reach values appropriate for a panel over simple supports. Vibration frequencies were determined by setting the air dynamic pressure and damping equal to zero and by searching for values of the frequency that satisfy the equations of equilibrium. Thirteen stations between supports were deemed sufficient and were used in the calculations for vibration frequency and flutter pressure. Flutter pressures were determined, as indicated in appendix A, by examining the behavior of the dynamic pressure with frequency. Typical dynamic-pressure—frequency curves are shown in figure 3 for selected values of support stiffness; considerable overlap of curves is indicated. This behavior was noted and discussed in reference 2 for a panel on inflexible multiple supports.

The dynamic pressure at flutter is given in figure 4 as a function of support stiffness. The flutter pressure is zero for zero support stiffness and generally increases with support stiffness, depending upon the aerodynamic damping (altitude). The curve for zero damping in figure 4 is based on steady-state aerodynamics, and the flutter dynamic pressure according to this curve drops sharply in the neighborhood of the stiffness at which the first and second natural frequencies (fig. 2) appear to be equal. In the neighborhood of this sharp drop, aerodynamic damping becomes important and modest amounts of damping wash out the dropoff. Presumably other stiffnesses for which the natural frequencies appear to be equal may have associated with them other sharp drops in flutter dynamic pressure for zero damping. It is to be expected that such drops would again be washed out by damping effects, and accordingly, they are not of concern.

Flutter modes were determined for the same four support-stiffness values considered in figure 3 and are presented in figure 5, where they exhibit remarkable change in character with change in support stiffness. Previous experience has indicated that flutter deflections are prominent in the trailing-edge bays. However, it can be seen from figure 5 that when the supports are weak, deflections are more prominent in the leading-edge bays.

Panels on Equally Spaced Supports

For steady-state theory the dynamic pressure at flutter was determined for various values of support stiffness and plotted in figure 6 for panels on two, three, four, or five equally spaced supports. For this case the outer supports are at the ends of the panel. Again 13 stations were placed between supports. The curves generally have the same shape as that discussed earlier for the special five-support configuration. The panel with two supports, one at each end of the panel, is an exception in that it exhibits no sharp dropoff. At higher values of support stiffness, the flutter dynamic pressure for each case stops increasing and approaches the corresponding value given in reference 2 for panels over nondeflecting (simple) supports. To compare the results for the panel on five equally spaced supports of figure 6 with the results for the special configuration of figure 4, the length of the panel of figure 4 should be changed by the factor 112/117 to account for the overhangs. The most interesting difference noted is that the dropoff extends over a larger range of support stiffness when there is no overhang. Examination of the results shows that the dropoff occurs when the support-stiffness parameter $ka^3/D(N-1)^3$ is about 100; this dropoff always occurred in the neighborhood of two equal eigenvalues. The maximum flutter pressure attainable occurs when the value of the support parameter $ka^3/D(N-1)^3$ is greater than 2000.

Panels on Unequally Spaced Supports

The dynamic pressure at flutter was determined for panels with simply supported ends and with an additional simple support at various locations. Flutter dynamic pressure is plotted in figure 7 as a function of the location of the support. The highest flutter pressure occurs when the support is at the center of the panel, but it drops off sharply, picks up again, then drops continuously as the support is moved toward either end of the panel. It should be mentioned that the flutter pressures presented in figure 7 were obtained from a finite-difference representation with 31 nodal points, and no attempt was made in this problem to find special behavior that might appear as the interval between nodal points is decreased.

CONCLUDING REMARKS

A method of analysis has been developed for the flutter behavior at supersonic speeds of wide panels with discrete flexible supports. Equations developed from wide-plate structural theory coupled with Ackeret aerodynamic theory are solved directly by means of finite differences. The listing is given of a computer program which is based on this approach and which is applicable to flutter of panels with thickness and mass variations along the length of the panel; panel support is represented by discrete extensional and rotational springs across the panel at various locations along the length of the panel.

Results are presented for three parameter studies. The first and second studies are concerned with the effect of support stiffness on panels with equally spaced supports. The panel flutter pressure generally increases with increasing support stiffness up to a limiting value, which occurs at values of the support-stiffness parameter $ka^3/D(N-1)^3$ above 2000, where N is the number of supports, k is the support stiffness per unit width, a is the panel length, and D is the plate bending stiffness. At values of support stiffness corresponding to two equal eigenvalues, $ka^3/D(N-1)^3$ of about 100, however, the flutter pressure decreases to a value that is highly dependent on the damping. For low damping (high altitude) this value may be so low that panel flutter poses a severe design constraint; thus, for low damping, such values of support stiffness should be avoided. The third study is concerned with an end-supported panel with an additional off-center support; it was found that the calculated flutter pressure varies in a nonuniform way with change in the location of the off-center support.

The method of analysis presented herein was found to be adequate for the problems treated. This demonstrates that direct techniques may be used for flutter analysis – that modal techniques are not necessarily required. The use of the computer program for these flutter problems requires considerable interface time on the part of the analyst. It is difficult to automate a computer program for flutter of a panel over multiple supports because of the multiplicity of eigenvalues in the critical range and because of the narrow ranges of support stiffness where damping is important.

Langley Research Center,
National Aeronautics and Space Administration,
Hampton, Va., December 6, 1973.

APPENDIX A

ANALYSIS

Difference equations will be derived to replace the differential equation of equilibrium and the boundary conditions, a method of solution will be discussed, and in appendix B a computer program will be presented which uses the method of solution to solve the difference equations.

Difference Equations

For simplicity the aerodynamic damping terms and rotational spring term are omitted from equation (1), which then becomes

$$\frac{\partial^2 \overline{M}_x}{\partial x^2} + \sum_{n=1}^{N} kw\delta(x-x_n) + m\frac{\partial^2 w}{\partial t^2} + \frac{2q}{\beta}\frac{\partial w}{\partial x} = 0 \tag{A1}$$

where

$$\overline{M}_x = D\frac{\partial^2 w}{\partial x^2} \tag{A2}$$

Periodic motion is assumed as

$$w = W(x)\sin \omega t \tag{A3}$$

so that equation (A1) becomes the following differential equation:

$$M_x'' + \sum_{n=1}^{N} kW\delta(x-x_n) - m\omega^2 W + \frac{2q}{\beta}W' = 0 \tag{A4}$$

with

$$M_x = DW'' \tag{A5}$$

Boundary conditions considered at $x = 0$ or a are

Free:

$$M_x = M_x' = 0 \tag{A6}$$

Simply supported:

$$W = M_x = 0 \tag{A7}$$

Clamped:

$$W = W' = 0 \tag{A8}$$

In addition, a spring could be located at $x = 0$ or a.

Using nodal points with spacing ϵ and conventional central differences gives equations (A4) and (A5) in finite-difference form as follows:

$$\begin{aligned} &\left(M_{x,j+1} - 2M_{x,j} + M_{x,j-1}\right)\frac{1}{\epsilon^2} + \sum_{n=1}^{N} k_n W_j \frac{1}{\epsilon} \delta_{jj_n} - m_j \omega^2 W_j \\ &+ E_j \frac{q}{\beta}\left(W_{j+1} - W_{j-1}\right)\frac{1}{\epsilon} = 0 \end{aligned} \tag{A9}$$

with

$$M_{x,j} = D_j\left(W_{j+1} - 2W_j + W_{j-1}\right)\frac{1}{\epsilon^2} \tag{A10}$$

where j_n is the value of j corresponding to the location of the nth spring and E_j is a coefficient (usually equal to 1) inserted for convenience in satisfying boundary conditions yet retaining the form of equilibrium equations (A9) and (A10) as required by the computer program. Besides retaining the form of equations (A9) and (A10), the boundary conditions are set up so that W_3 is the lowest subscript, nonzero value of W; and the first equation considered, therefore, is the one specifying equilibrium about the point corresponding to $j = 3$. Note that D, k, and m are now identified with subscripts, so that stiffness (or thickness) and mass variation may be permitted and the various spring supports may have different stiffnesses as well as different locations.

For a spring-supported edge at $x = 0$ or $j = 4$, the boundary conditions from equation (A6), including a term for the spring, become

$$M_{x,4} = 0 \tag{A11}$$

$$\left(M_{x,5} - M_{x,3}\right)\frac{1}{2\epsilon} + k_0 W_4 = 0 \tag{A12}$$

After substitution from equation (A12), equation (A9) representing equilibrium at $j = 4$ becomes

$$\left(2M_{x,5} - 2M_{x,4}\right)\frac{1}{\epsilon^2} + 2k_0W_4\,\frac{1}{\epsilon} - M_4\omega^2W_4 + E_4\,\frac{q}{\beta}\left(W_5 - W_3\right)\frac{1}{\epsilon} = 0 \tag{A13}$$

The form of equations (A9) and (A10) may be retained and yet the boundary conditions of equations (A11) and (A13) are satisfied if W_3, D_4, m_4, and E_4 are the first nonzero values of these parameters according to subscript, D_4 and m_4 are one-half their corresponding interior value, and $E_4 = \frac{1}{2}$. The initial equation to be considered corresponds to $j = 3$. In the interior, the values of E_j are always unity. For a free edge at $j = 4$, simply set $k_0 = 0$.

For a free edge at $x = 0$, halfway between $j = 3$ and $j = 4$ the boundary conditions from equation (A6) become

$$\frac{M_{x,3} + M_{x,4}}{2} = 0 \tag{A14}$$

$$\frac{M_{x,3} - M_{x,4}}{\epsilon} = 0 \tag{A15}$$

or $M_{x,3} = M_{x,4} = 0$. For this condition, the form of equations (A9) and (A10) is retained by starting with the same values as before except that D_4, m_4, and E_4 are the same as their corresponding interior values.

For a simply supported edge at $x = 0$ or $j = 2$, the boundary conditions from equation (A7) are

$$W_2 = M_{x,2} = 0 \tag{A16}$$

The form of equations (A9) and (A10) is retained for these conditions if the first nonzero values of the parameters according to subscript have subscript 3 and the initial equation corresponds to $j = 3$.

For a clamped edge at $x = 0$ or $j = 2$, the boundary conditions from equation (A8) are

$$W_2 = 0 \tag{A17}$$

$$W_3 - W_1 = 0 \tag{A18}$$

Thus,

$$M_{x,2} = \frac{2D_2}{\epsilon^2} W_3 \tag{A19}$$

To retain the form of equations (A9) and (A10), the first nonzero values of the parameters according to subscript are W_3, D_2, m_3, and E_3, and the initial equation corresponds to $j = 3$. The value taken for D_2 must be twice its corresponding interior value. Similar treatment must be given to boundary conditions at $x = a$.

Method of Solution

The difference equations just presented are linear and homogeneous (right-hand side = 0) and each has a maximum of five unknown deflections W_j; the values of j that appear in each equation are adjacent, that is, $j + 2$, $j + 1$, j, $j - 1$, and $j - 2$. The equation can be arranged so that the first and last equations have only three unknowns and the second and next to last equations have only four unknowns. When written in matrix form, they can be identified as being banded, with a bandwidth of 5. A modal solution would start out by setting $q = 0$ and then solving for the modes and frequencies for which the determinant of the coefficients vanishes. The modes and frequencies would then be used in a separate flutter analysis. A direct solution would usually be obtained by finding values of q and ω for which the determinant of the coefficients vanishes. The present method of solution is direct, but instead of working with the determinant of the coefficients of the banded matrix, a marching procedure is used.

If the first two unknowns $(W_3$ and $W_4)$ are assumed, the first equation will determine the third unknown, the second equation will determine the fourth unknown, and so forth, and all the unknowns would be determined without using the last two equations. The present method of solution for given values of q and ω assumes two linearly independent sets of values for the first and second unknowns, and thus, two sets of preliminary solutions are obtained on the basis of these assumptions. Then a linear combination of the preliminary solutions is substituted into the last two equations and the determinant of the coefficients of these two equations is found. A solution curve is obtained by finding combinations of q and ω for which this determinant is zero.

The standard coalescence approach for undamped flutter was used in this method of solution in that extremum values of q along the q-ω solution curves correspond to flutter values (fig. 3), with the lowest of such values determining the flutter pressure of interest. For the damped case, the solution points where the real and imaginary parts of the determinant vanish simultaneously correspond to flutter values.

APPENDIX B

COMPUTER PROGRAM

The computer program based on the equations just derived in appendix A applies to wide panels of any prescribed stiffness distribution, any number and location of discrete flexible supports, any number and location of discrete rotational supports, and any mass distribution.

Because of the intersecting solution curves that occur for the panels of interest, it is difficult to automate the solution for the flutter pressure. However, the computer program has options to determine q for a set of ω, ω for a set of q, the value of the determinant for each ω-q combination, and for this option, an estimate of ω when there is a determinant crossing at a given value of q.

```
      PROGRAM SUPRFLX(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C     ***
C     SUPERSONIC FLUTTER OF PANELS ON FLEXIBLE SUPPORTS
C     ***
C
C     EI    PANEL STIFFNESS/REFERENCE STIFFNESS(D)
C
C     EM    PANEL MASS/REFERENCE MASS(M)
C
C     EK    SUPPORT EXTENSIONAL STIFFNESS,K*A**3/D*EL**3
C
C     C     SUPPORT ROTATIONAL STIFFNESS,C*A/D*EL
C
C     W     DEFLECTION
C
C     EQ    Q TERM COEFFICIENT
C
C     IOPT=1 DETERMINE DET FOR(OMEGA,Q)
C         =2 DETERMINE Q FOR EACH OMEGA
C         =3 DETERMINE OMEGA FOR EACH Q
C
C     NDS   NUMBER OF NONZERO W
C
C     EL    A/EPSILON
C
C     OMEGA       FREQUENCY,OMEGA*SQRT(M*A**4/D)
C
C     Q           FLUTTER PRESSURE,2.*Q*A**3/BETA/D
C
C     DET         DETERMINANT COMPUTED
C
C     *                                                                     *
C     *****      *****      *****      *****      *****      *****      ******
      DIMENSION EI(61),EM(61),EK(61),C(61),W(61),X(3),Y(3),CASE(8)
      DIMENSION OMEG(50),DETT(50),EQ(61)
      NAMELIST/OPTION/IOPT,NDS,EL
      NAMELIST/FLUTER/OMEGA1,DELOM,OMFIN,Q,DELQ,QFIN
      NAMELIST/STIFF/EI,EK,C,EM,EQ
C
C         BILINEAR APPROXIMATION FOR THE VALUE OF X AT Y=0
```

APPENDIX B

```
      ABSCISA(X3,X2,X1,Y3,Y2,Y1)=X3+((Y1*Y3-Y2*Y3)*(X3-X1)*(X2-X3))/(Y1*
     1Y2*(X1-X2)+Y1*Y3*(X3-X1)+Y2*Y3*(X2-X3))
    1 READ (5,5001) CASE
      IF(EOF,5) 4,5
    4 PRINT 8997
 8997 FORMAT(//*                    STOPPED ON EOF,5                   *)
      STOP
    5 WRITE (6,5000) CASE
      READ OPTION
      READ FLUTER
      WRITE(6,OPTION)
      WRITE(6,FLUTER)
      NDEE=NDS +4
      READ STIFF
      OMEGA1=OMEGA1/EL**2
      OMEGA=OMEGA1
      DELOM=DELOM/EL **2
      OMFIN=OMFIN/EL **2
      Q=Q/EL**3/2.
      DELQ=DELQ/EL**3/2.
      QFIN=QFIN/EL**3/2.
 5000 FORMAT(////15H INPUT FOR CASE///1X8A10)
 5001 FORMAT(8A10)
      WRITE(6,5009)(I,EI(I),EM(I),EK(I),C(I),EQ(I),I=1,NDEE)
 5009 FORMAT(/5H0  I   6X5HEI(I)12X5HEM(I)11X5HEK(I)12X4HC(I)11X5HEQ(I)
     1 /(I5,5E16.8))
      KOUNT=0
      WRITE(6,5010)
 5010 FORMAT(/9X5HOMEGA16X1HQ19X3HDET24X14HNO. ITERATIONS/)
      IF(IOPT.EQ.1)GO TO 100
      IF(IOPT.EQ.3) GO TO 300

C          DETERMINE Q FOR EACH OMEGA

   30 KOUNT=0
      HOLD=Q
      DO 40 I=1,3
      IF(I.GE.2) Q=Q+DELQ
      CALL DETERM(OMEGA,Q,EM,EI,NDS,        DET,C,EK,W,EQ)
      X(I)=DET
      Y(I)=Q
   40 CONTINUE
   50 CISSA=ABSCISA(Y(3),Y(2),Y(1),X(3),X(2),X(1))
      Q=CISSA
      IF(KOUNT.EQ.0)X0=X(1)
      KOUNT=KOUNT+1
      CALL DETERM(OMEGA,Q,EM,EI,NDS,        DET,C,EK,W,EQ)
      X(1)=DET
      Y(1)=Q
      IF(ABS(X(1))/ABS(X0).LE.1.E-5) GO TO 80
      IF(KOUNT.GE.20) GO TO 70
      DELQ=ABS((Q-HOLD)/10.)
      DO 60 I=2,3
      Q=Q+DELQ
      CALL DETERM(OMEGA,Q,EM,EI,NDS,        DET,C,EK,W,EQ)
      X(I)=DET
      Y(I)=Q
   60 CONTINUE
      GO TO 50
```

APPENDIX B

```
 70   O=OMEGA*EL**2
      QQ=Q*EL**3*2.
      WRITE(6,250)O,QQ,DET,KOUNT
      WRITE(6,260)
      GO TO 1
 80   O=OMEGA*EL**2
      QQ=Q*EL**3*2.
      WRITE(6,250)O,QQ,DET,KOUNT
      OMEGA=OMEGA+DELOM
      IF(OMEGA.GT.OMFIN)GO TO 1
      GO TO 30

C          DETERMINE DET FOR EACH (OMEGA,Q)

100   L=0
101   L=L+1
      CALL DETERM(OMEGA,Q,EM,EI,NDS,        DET,C,EK,W,EQ)
      OMEG(L)=OMEGA*EL**2
      DETT(L)=DET
      QQ=Q*EL**3*2.
      WRITE(6,250)OMEG(L),QQ,DETT(L),KOUNT
      IF(L.LT.3)GO TO 102

C          IDENTIFY CHANGE IN SIGN

      D1=DETT(L)*DETT(L-1)
      D2=DETT(L)*DETT(L-2)
      IF(D1.GE.0.AND.D2.GE.0) GO TO 102

C          INTERPOLATE

103   CISSA=ABSCISA(OMEG(L),OMEG(L-1),OMEG(L-2),DETT(L),DETT(L-1),
     *DETT(L-2))
      O=CISSA
      DET=0.
      WRITE(6,250)O,QQ,DET,KOUNT
102   PRINT 120
      OMEGA=OMEGA+DELOM
      IF(OMEGA.GT.OMFIN)GO TO 110
      GO TO 101
110   WRITE(6,120)
120   FORMAT(   )
      Q=Q+DELQ
      OMEGA=OMEGA1
      IF(Q.GT.QFIN)GO TO 1
      GO TO 100

C          DETERMINE OMEGA FOR EACH Q

300   KOUNT=0
      DO 310 I=1,3
      IF(I.GE.2) OMEGA=OMEGA+DELOM
      CALL DETERM(OMEGA,Q,EM,EI,NDS,        DET,C,EK,W,EQ)
      X(I)=DET
      Y(I)=OMEGA
310   CONTINUE
```

APPENDIX B

```
320   CISSA=ABSCISA(Y(3),Y(2),Y(1),X(3),X(2),X(1))
      OMEGA=CISSA
      IF(KOUNT.EQ.0) XO=X(1)
      KOUNT=KOUNT+1
      CALL DETERM(OMEGA,Q,EM,EI,NDS,        DET,C,EK,W,EQ)
      X(1)=DET
      Y(I)=OMEGA
      IF(KOUNT.GE.10)  GO TO 70
      IF(ABS(X(1))/ABS(XO).LE.1.E-5) GO TO 340
      DELOM=DELOM/10.
      DO 330 I=2,3
      OMEGA =OMEGA+DELOM
      CALL DETERM(OMEGA,Q,EM,EI,NDS,        DET,C,EK,W,EQ)
      X(I)=DET
      Y(I)=OMEGA
330   CONTINUE
      GO TO 320
340   O=OMEGA*EL**2
      QQ=Q*EL**3*2.
      PRINT 250, O,QQ,DET,KOUNT
      WRITE(6,250)O,QQ,DET,KOUNT
      Q=Q+DELQ
      IF(Q.GT.QFIN) GO TO 1
      GO TO 300
250   FORMAT(/3E20.8,12XI5)
260   FORMAT(1X29H10 ITERATIONS, NO CONVERGENCE)
      END
```

APPENDIX B

```
      SUBROUTINE DETERM(OMEGA,Q,EM,EI,NDS,DET,C,EK,W,EQ)
C          EVALUATE DETERMINANT FOR (OMEGA,Q)
      DIMENSION EM(1),EI(1),W(1),      E(2,2),C(1),EK(1)
      DIMENSION EQ(1)
C          WIDE PLATE EQUATIONS IN DIFFERENCE FORM
      WXX(K)=W(K+1)-2.*W(K)+W(K-1)
      XM(K)= EI(K)*WXX(K)
      XEM(K)=XM(K+1)-2.*XM(K)+XM(K-1)+Q*EQ(K)*(W(K+1)-W(K-1))
     1       -C(K+1)*(W(K+1)-W(K))+C(K)*(W(K)-W(K-1))
     2       +(EK(K)-OMEGA**2*EM(K))*W(K)
      NDSS=NDS+1
      NDSE=NDS+2
      NDEE=NDS+4
      DO 60 I=1,NDEE
C          MARCHING PROCEDURE
   60 W(I)=0.0
      DO 90 N=3,4
      MN=0
      W(3)=0.
      W(4)=0.
      W(N)=1.
      DO 70 J=3,NDS
      EXM=2.*XM(J)-XM(J-1)-Q*EQ(J)*(W(J+1)-W(J-1))+C(J+1)
     1    *(W(J+1)-W(J))-C(J)*(W(J)-W(J-1))-(EK(J)-OMEGA**2*EM(J))*W(J)
      EXX= EXM/EI(J+1)
      W(J+2)=EXX+2.*W(J+1)-W(J)
   70 CONTINUE
C          COEFFICIENTS OF LAST TWO EQUATIONS
      DO 80 J=NDSS,NDSE
      MN=MN+1
      E(MN,N-2)=XEM(J)
   80 CONTINUE
   90 CONTINUE
C          DETERMINANT OF COEFFICIENTS
      T1=E(1,1)-E(2,1)
      T2=E(2,2)-E(1,2)
      T3=T1*E(1,2)+T2*E(2,1)
      DET=T1*T2+T3
      RETURN
      END
```

REFERENCES

1. Anon.: Panel Flutter. NASA Space Vehicle Design Criteria (Structures). NASA SP-8004, 1964. (Revised 1972.)
2. Dowell, Earl: Flutter of Multibay Panels at High Supersonic Speeds. AIAA J., vol. 2, no. 10, Oct. 1964, pp. 1805-1814.
3. Hedgepeth, John M.: Flutter of Rectangular Simply Supported Panels at High Supersonic Speeds. J. Aeronaut. Sci., vol. 24, no. 8, Aug. 1957, pp. 563-573, 586.



L-71-5508

Figure 1.- Panel on standoff supports considered for the thermal protection system of a reentry vehicle.



Figure 2.- Lowest natural vibration frequencies of the special panel on multiple flexible supports.



Figure 3.- Typical curves from which (flutter) values of q are determined at frequency coalescence. (The symbol × denotes flutter.)



Figure 4.- Flutter of the special panel on multiple flexible supports.



Figure 5.- Effect of support stiffness on the flutter mode of the special panel.



Figure 6.- Flutter of panels on equally spaced flexible supports. (Asymptotic values were taken from ref. 2.)



Figure 7.- Flutter of panels on unequally spaced supports.

NASA-Langley, 1974 L-9206

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TN D-7443		

4. Title and Subtitle

FLUTTER OF PANELS ON DISCRETE FLEXIBLE SUPPORTS

5. Report Date

March 1974

6. Performing Organization Code

7. Author(s)

Manuel Stein

8. Performing Organization Report No.

L-9205

10. Work Unit No.

502-22-10-01

9. Performing Organization Name and Address

NASA Langley Research Center
Hampton, Va. 23665

11. Contract or Grant No.

13. Type of Report and Period Covered

Technical Note

12. Sponsoring Agency Name and Address

National Aeronautics and Space Administration
Washington, D.C. 20546

14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

The supersonic flutter of wide panels on discrete flexible supports is investigated for three different panel-support configurations. The first study examines the effect of support stiffness on the flutter behavior ot a specific five-support configuration with leading- and trailing-edge overhangs; this configuration was recently under consideration for a heat shield on an entry body. The second study investigates the effect of support stiffness on flutter of panels with various numbers of equally spaced supports. The third study examines the effect of center-support location on the flutter of panels with three supports. Results are presented in nondimensional form. The analysis is based on wide-plate structural theory and Ackeret aerodynamics. Finite differences are employed to obtain solutions for flutter pressure. A computer program based on this analysis and including a direct solution technique is presented. The program can be used to find the flutter pressure of wide panels of variable thickness supported by any number of flexible supports.

17. Key Words (Suggested by Author(s))

Flutter
Panel flutter
Spring-supported panel

18. Distribution Statement

Unclassified – Unlimited

STAR Category 32

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	22	$2.75

*For sale by the National Technical Information Service, Springfield, Virginia 22151

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
WASHINGTON, D.C. 20546

OFFICIAL BUSINESS
PENALTY FOR PRIVATE USE $300

SPECIAL FOURTH-CLASS RATE
BOOK

POSTAGE AND FEES PAID
NATIONAL AERONAUTICS AND
SPACE ADMINISTRATION
451

U.S.MAIL

POSTMASTER: If Undeliverable (Section 158 Postal Manual) Do Not Return

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—NATIONAL AERONAUTICS AND SPACE ACT OF 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons. Also includes conference proceedings with either limited or unlimited distribution.

CONTRACTOR REPORTS: Scientific and technical information generated under a NASA contract or grant and considered an important contribution to existing knowledge.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities. Publications include final reports of major projects, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

TECHNOLOGY UTILIZATION PUBLICATIONS: Information on technology used by NASA that may be of particular interest in commercial and other non-aerospace applications. Publications include Tech Briefs, Technology Utilization Reports and Technology Surveys.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION OFFICE

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546